

# Lección 2

Marcos Bujosa

17 de septiembre de 2024

## Índice

<b>1. Componentes no observables del número de viajeros internacionales</b>	<b>1</b>
1.1. Actividad 1 - generar una serie con el índice temporal y su cuadrado . . . . .	1
1.2. Actividad 2 - Ajustar una tendencia lineal . . . . .	2
1.3. Actividad 3 - Ajustar una tendencia cuadrática . . . . .	3
1.4. Actividad 4 - Ajustar una tendencia cuadrática y un componente estacional determinista . . . . .	3
<b>2. Componentes no observables del número de viajeros internacionales 2</b>	<b>6</b>
2.1. Actividad 1 - Estime el modelo con tendencia cuadrática y estacionalidad determinista . . . . .	6
2.2. Actividad 2 - Reducir el modelo eliminando secuencialmente variables no significativas . . . . .	6
2.3. Actividad 3 - Contrastar la ausencia de autocorrelación . . . . .	6
2.4. Actividad 4 - Estimación del modelo con errores estándar robustos . . . . .	7
2.5. Actividad 4 - Estimación incluyendo en el modelo la autocorrelación de orden uno en las perturbaciones . . . . .	7

## 1. Componentes no observables del número de viajeros internacionales

Guión: `componentesAirlinePass.inp`

Vamos a reproducir el ejemplo de estimación de componentes no observables visto en clase. Los datos son mensuales y contienen el número de viajeros internacionales (en miles) en las líneas aéreas norteamericanas en los años 1949–1960 que aparece en manual de Box & Jenkins.

### 1. Objetivo

a) Reproducir el ejemplo visto en la lección 2.

### 2. Carga de datos *Archivo -->Abrir datos -->Archivo de muestra* y en la pestaña *Gretl* seleccione *bjg*.

*o bien teclee en línea de comandos:*

```
open bjg
```

### 1.1. Actividad 1 - generar una serie con el índice temporal y su cuadrado

Pulse en el menú desplegable *Añadir* que aparece arriba, en el centro de la ventana principal de *Gretl*.

■ *Añadir ->Variable índices*

*o bien teclee en línea de comandos:*

```
genr time
```

Seleccione con el ratón la variable `time` y luego pulse en el menú desplegable *Añadir* que aparece arriba, en el centro de la ventana principal de *Gretl*.

■ *Añadir ->Cuadrados de las variables seleccionadas*

*o bien teclee en línea de comandos:*

```
square time
```

## 1.2. Actividad 2 - Ajustar una tendencia lineal

- Estime el modelo mediante los menús desplegables: *Modelo ->Mínimos Cuadrados Ordinarios*; indique a *Gretl* el regresando y los regresores y pulse *Aceptar*.

- “Pinche” *Archivo* en la ventana del modelo estimado y seleccione *guardar como un icono y cerrar* o bien teclee en línea de comandos:

```
TendenciaLineal <- ols lg 0 time
```

(el cero 0 indica el término constante: const)

- Recupere el modelo “pinchando” sobre su icono

o teclee en línea de comandos el nombre que ha dado al icono seguido de *.show*, es decir:

```
TendenciaLineal.show
```

### 1. Recuperemos los valores ajustados

- Desde la ventana del modelo ajustado (recupérese con su icono), “pinche” en *guardar ->valores estimados*. Elija como nombre *phat* (puede añadir una descripción de la variable). Pulse en *Aceptar*
- Repita para guardar los **residuos** con el nombre *ehat*

o bien teclee en línea de comandos:

```
series TendenciaLineal = $yhat  
series Comp_irregular = $uhat
```

### 2. Gráfico de la serie y la tendencia lineal

- Marque las variables *lg* y *Tendencia* (pulsando *ctrl*). Pinche con el botón derecho del ratón sobre ellas. Elija *Gráfico de series temporales*

o bien teclee en línea de comandos:

```
GraficoTendenciaLineal <- gnuplot lg TendenciaLineal --time-series --with-lines  
GraficoTendenciaLineal.show
```

En la zona inferior izquierda de la ventana principal puede ver una serie de iconos. Uno de ellos es la *vista de iconos de sesión*.

### 3. Gráfico del componente irregular

- Marque la variable correspondiente al componente irregular (pulsando *ctrl*) y pinche con el botón derecho del ratón sobre ella. Elija *Gráfico de series temporales*

o bien teclee en línea de comandos:

```
GraficoComponenteIrregular <- gnuplot Comp_irregular --time-series --with-lines  
GraficoComponenteIrregular.show
```

### 4. Gráfico de la diferencia estacional del componente irregular

- Seleccione con el ratón la variable correspondiente al componente irregular y luego pulse en el menú desplegable *Añadir* que aparece arriba, en el centro de la ventana principal de *Gretl*.

- *Añadir ->Diferencias estacionales de las variables seleccionadas*

o bien teclee en línea de comandos:

```
sdiff Comp_irregular
```

- Genere un gráfico con la nueva serie diferenciada estacionalmente

en línea de comandos:

```
GraficoComponenteIrregularD12 <- gnuplot sd_Comp_irregular --time-series --with-lines  
GraficoComponenteIrregularD12.show
```

### 1.3. Actividad 3 - Ajustar una tendencia cuadrática

Repita el ejercicio anterior, pero ajustando una tendencia cuadrática

- Estime la tendencia por MCO y vea los resultados de la regresión
- Guarde los valores ajustados (TendenciaCuadratica)
- Guarde los residuos (ComponenteIrregular2)
- Dibuje la tendencia
- Dibuje el componente irregular
- Dibuje la diferencia estacional del componente irregular

### 1.4. Actividad 4 - Ajustar una tendencia cuadrática y un componente estacional determinista

Pulse en el menú desplegable *Añadir* que aparece arriba, en el centro de la ventana principal de [Gretl](#).

- *Añadir -> Variables ficticias estacionales*

*o bien teclee en línea de comandos:*

```
seasonals()
```

- Estime el modelo mediante los menús desplegables: *Modelo -> Mínimos Cuadrados Ordinarios*; indique a [Gretl](#) el regresando y los regresores y pulse *Aceptar*.

- “Pinche” *Archivo* en la ventana del modelo estimado y seleccione *guardar como un icono y cerrar*

*o bien teclee en línea de comandos:*

```
ModeloCompleto <- ols lg const time sq_time S1 S2 S3 S4 S5 S6 S7 S8 S9 S10 S11
ModeloCompleto.show
```

#### 1. Genere una nueva serie con la tendencia y otra con el componente estacional estimados

- Cálculo de la tendencia estimada:  $\hat{\beta}_1 \text{const} + \hat{\beta}_2 \text{time} + \hat{\beta}_3 \text{sq\_time}$   
*Guardar -> Definir una nueva variable* y teclee:

```
series Tendencia3 = $coeff[1] + $coeff[2]*time + $coeff[3]*sq_time
```

o bien:

```
series Tendencia3 = $coeff(const) + $coeff(time)*time + $coeff(sq_time)*sq_time
```

- De manera análoga genere una serie con el componente estacional

```
series Comp_Estacional3 = $coeff(S1)*S1 + $coeff(S2)*S2 + $coeff(S3)*S3 + $coeff(S4)*S4 \
+ $coeff(S5)*S5 + $coeff(S6)*S6 + $coeff(S7)*S7 + $coeff(S8)*S8 \
+ $coeff(S9)*S9 + $coeff(S10)*S10 + $coeff(S11)*S11
```

- Genere los siguientes gráficos
  - la serie y su tendencia cuadrática
  - El componente estacional
  - El componente irregular
  - La serie y su ajuste

```

GraficoTendencia3 <- gnuplot lg Tendencia3 --time-series --with-lines
GraficoTendencia3.show

GraficoComponenteEstacional3 <- gnuplot Comp_Estacional3 --time-series --with-lines
GraficoComponenteEstacional3.show

series ComponenteIrregular3 = $uhat
GraficoComponenteIrregular3 <- gnuplot ComponenteIrregular3 --time-series --with-lines
GraficoComponenteIrregular3.show

series Ajuste3 = $yhat
GraficoAjuste3 <- gnuplot lg Ajuste3 --time-series --with-lines
GraficoAjuste3.show

```

## Código completo de la práctica componentesAirlinePass.inp

```

open bjpg
genr time
square time

TendenciaLineal <- ols lg 0 time
TendenciaLineal.show

series TendenciaLineal = $yhat
series Comp_irregular = $uhat

GraficoTendenciaLineal <- gnuplot lg TendenciaLineal --time-series --with-lines
GraficoTendenciaLineal.show

GraficoComponenteIrregular <- gnuplot Comp_irregular --time-series --with-lines
GraficoComponenteIrregular.show

sdiff Comp_irregular

GraficoComponenteIrregularD12 <- gnuplot sd_Comp_irregular --time-series --with-lines
GraficoComponenteIrregularD12.show

# modelo con tendencia cuadrática
TendenciaCuadratica <- ols lg 0 time sq_time
TendenciaCuadratica.show

series TendenciaCuadratica = $yhat
series Comp_irregular2 = $uhat

GraficoTendenciaCuadratica <- gnuplot lg TendenciaCuadratica --time-series --with-lines
GraficoTendenciaCuadratica.show

GraficoComponenteIrregular2 <- gnuplot Comp_irregular2 --time-series --with-lines
GraficoComponenteIrregular2.show

sdiff Comp_irregular2

GraficoComponenteIrregular2D12 <- gnuplot sd_Comp_irregular2 --time-series --with-lines
GraficoComponenteIrregular2D12.show

# modelo con tendencia cuadrática

seasonals()

ModeloCompleto <- ols lg const time sq_time S1 S2 S3 S4 S5 S6 S7 S8 S9 S10 S11
ModeloCompleto.show

```

```

series Tendencia3 = $coeff[1] + $coeff[2]*time + $coeff[3]*sq_time

series Tendencia3 = $coeff(const) + $coeff(time)*time + $coeff(sq_time)*sq_time

series Comp_Estacional3 = $coeff(S1)*S1 + $coeff(S2)*S2 + $coeff(S3)*S3 + $coeff(S4)*S4 \
+ $coeff(S5)*S5 + $coeff(S6)*S6 + $coeff(S7)*S7 + $coeff(S8)*S8 \
+ $coeff(S9)*S9 + $coeff(S10)*S10 + $coeff(S11)*S11

GraficoTendencia3 <- gnuplot lg Tendencia3 --time-series --with-lines
GraficoTendencia3.show

GraficoComponenteEstacional3 <- gnuplot Comp_Estacional3 --time-series --with-lines
GraficoComponenteEstacional3.show

series ComponenteIrregular3 = $uhat
GraficoComponenteIrregular3 <- gnuplot ComponenteIrregular3 --time-series --with-lines
GraficoComponenteIrregular3.show

series Ajuste3 = $yhat
GraficoAjuste3 <- gnuplot lg Ajuste3 --time-series --with-lines
GraficoAjuste3.show

```

## 2. Componentes no observables del número de viajeros internacionales 2

Guión: [componentesAirlinePass2.inp](#)

Continuamos el ejemplo anterior, pero ahora vamos a reducir el modelo quitando variables no significativas.

### 2.1. Actividad 1 - Estime el modelo con tendencia cuadrática y estacionalidad determinista

Repita la estimación del último modelo de la practica anterior

```
open bjt
genr time
square time
seasonals()
ModeloInicial <- ols lg const time sq_time S1 S2 S3 S4 S5 S6 S7 S8 S9 S10 S11
ModeloInicial.show
```

### 2.2. Actividad 2 - Reducir el modelo eliminando secuencialmente variables no significativas

Los p-valores de algunos parámetros indican que sus estimaciones son no significativas. En particular los correspondientes a las variables ficticias de enero, febrero y octubre.

Reduzca el modelo, eliminando aquellas variables no significativas al 5%. Verifique que el conjunto de variables excluidas es conjuntamente no significativo.

- Desde la ventana del modelo estimado “pinche” en *contrast* -->*omitir variables* y marque la opción *Eliminación secuencial de variables utilizando el valor p a dos colas*, indique una significación del 5% y pulse en *Aceptar*

```
PrimeraReduccion <- omit --auto=0.05
```

### 2.3. Actividad 3 - Contrastar la ausencia de autocorrelación

- Observe bajo el valor de contraste de Durbin-Watson (0,691477).
- Desde la ventana del modelo estimado “pinche” en *contrast* -->*valor p del estadístico Durbin-Watson*. o bien teclee en línea de comandos:

```
scalar DW = $dw
scalar PDW = $dwpval
print DW
print PDW
```

Claramente se rechaza la ausencia de autocorrelación de orden uno.

- Desde la ventana del modelo estimado “pinche” en *contrast* -->*Autocorrelación*. E indique por ejemplo 3 retardos. o bien teclee en línea de comandos:

```
modtest --autocorr 3
```

Claramente se rechaza la ausencia de autocorrelación y se observa que el retardo de orden uno es muy significativo.

## 2.4. Actividad 4 - Estimación del modelo con errores estándar robustos

Los test de autocorrelación indican que la inferencia empleada para reducir el modelo es incorrecta. A la hora de calcular las desviaciones típicas de las estimaciones hay que tener en cuenta que las perturbaciones están autocorreladas.

- Estime el modelo inicial con errores estándar robustos: *Modelo ->Mínimos Cuadrados Ordinarios*; indique a *Gretl* el regresando y los regresores; marque la opción *Desviaciones típicas robustas* y pulse *Aceptar*.  
o bien teclee en línea de comandos:

```
ModeloInicialDTR <- ols lg const time sq_time S1 S2 S3 S4 S5 S6 S7 S8 S9 S10 S11 --robust
ModeloInicialDTR.show
```

Fíjese que al 5% de significación, solo es no significativa al dummie correspondiente al mes de febrero.

- Reduzca el modelo, eliminando aquellas variables no significativas al 5%. Verifique que el conjunto de variables excluidas es conjuntamente no significativo.

Desde la ventana del modelo estimado “pinche” en *contrastos -->omitir variables* y marque la opción *Eliminación secuencial de variables utilizando el valor p a dos colas*, indique una significación del 5% y pulse en *Aceptar* o bien teclee en línea de comandos:

```
ModeloReducidoDTR <- omit --auto=0.05
```

Fíjese que únicamente se elimina la dummie correspondiente a febrero.

## 2.5. Actividad 4 - Estimación incluyendo en el modelo la autocorrelación de orden uno en las perturbaciones

- Re-estime el modelo incluyendo en el modelo un AR(1) para las perturbaciones: *Modelo ->Series temporales univariantes ->Errores AR ->AR(1)* y pulse *Aceptar*.  
o bien teclee en línea de comandos:

```
ModeloAR1 <- ar1 lg const time sq_time S1 S2 S3 S4 S5 S6 S7 S8 S9 S10 S11
ModeloAR1.show
```

- Elimine secuencialmente las variables no significativas al 5%
- Haga un gráfico de los residuos y observe que "son estacionarios"(es decir, que tienen el aspecto de una realización de un proceso estacionario)
- Haga un gráfico de frecuencias de los residuos y observe que tiene la forma campaniforme compatible con una distribución gaussiana.

Marque con el ratón la variable **Residuos** y pinchado en la serie marcada con el botón derecho del ratón seleccione *Distribución de frecuencias*.

O bien mediante el comando

```
freq Residuos --show-plot
```

donde `--show-plot` indica que se genere el gráfico en una ventana. Observe que dicho comando también genera una ventana de texto con la distribución de frecuencias relativa y acumulada.

- Realice el contraste de normalidad de los residuos: Desde la ventana del modelo estimado “pinche” en *contrastos -->Normalidad de los residuos* y marque la opción *Eliminación secuencial de variables utilizando el valor p a dos colas*, indique una significación del 5% y pulse en *Aceptar*

O bien desde la ventana principal: marque la variable **Residuos** y “pinche” en *Variable -->Contraste de Normalidad*

O bien mediante el comando

```
normtest Residuos --all
```

## Código completo de la práctica componentesAirlinePass2.inp

```
open bjpg
genr time
square time
seasonals()
ModeloInicial <- ols lg const time sq_time S1 S2 S3 S4 S5 S6 S7 S8 S9 S10 S11
ModeloInicial.show

PrimeraReduccion <- omit --auto=0.05

scalar DW = $dw
scalar PDW = $dwpval
print DW
print PDW

modtest --autocorr 3

ModeloInicialDTR <- ols lg const time sq_time S1 S2 S3 S4 S5 S6 S7 S8 S9 S10 S11 --robust
ModeloInicialDTR.show

ModeloReducidoDTR <- omit --auto=0.05

ModeloAR1 <- ar1 lg const time sq_time S1 S2 S3 S4 S5 S6 S7 S8 S9 S10 S11
ModeloAR1.show

ModeloAR1Reducido <- omit --auto=0.05
ModeloAR1Reducido.show

series Residuos = $uhat
GraficoResiduos <- gnuplot Residuos --time-series --with-lines
GraficoResiduos.show

freq Residuos --show-plot

series Residuos = $uhat
GraficoResiduos <- gnuplot Residuos --time-series --with-lines
GraficoResiduos.show

normtest Residuos --all
```